

## Exercice 3 :

```

Program exe3_158 ;
Uses Wincrt;
Var ch1, ch2: String;
(***** p.p *****)
Function Sans_Redondance(ch1,ch2:string):string;
Var ch3, aux:string; i:integer;
Begin
  if length(ch1)>length(ch2)
  then begin
    aux:=ch1;
    ch1:=ch2;
    ch2:=aux;
  end;
  ch3:='';
  FOR i:=1To Length(ch1) Do
    IF (POS(ch1[i],ch2)<>0) and (POS(ch1[i],ch3)=0)
    Then ch3:=ch3+ch1[i];
  Sans_redondance:=ch3;
End;
(***** p.p *****)
Begin
  Write('Chaîne 1 = ');Readln (ch1);
  Write('Chaîne 2 = ');Readln (ch2);
  Writeln('Résultat : ', sans_redondance(ch1,ch2));
End.

```

## Exercice 4 :

```

Program exe4_158; {calcul sin(1)}
uses wincrt;
var n:integer;
(***** p.p *****)
function somme(n:integer):real;
var s:real;
  i,signe:integer;
{=====}
function fact(n:integer):longint;
var i:integer;
  f:longint;
begin
  f:=1;
  for i:=2 to n do f:=f*i;
  fact:=f;
end;
{=====}
begin
  s:=1; i:=1; signe:=1;
  repeat
    i:=i+2;
    signe:=-signe;
    s:=s+signe/fact(i);
  until i>=n;
  somme:=s;
end;
(***** p.p *****)
begin
  repeat
    write ('N = '); readln(n);
  until n in [5..30];
  writeln (somme(n):2:8);
end.

```

## Exercice 7 :

```

Program exe7_158;
uses wincrt;
var a:real;
(***** p.p *****)
function suite_heron(a:real):real;
const n=50;
var x:real;
  i:integer;
begin
  x:=a/2; {x0 choix arbitraire}
  for i:=1 to n do x:=(x + a/x)/2;
  suite_heron:=x;
end;
(***** p.p *****)
begin
  write('a = '); readln(a);
  writeln('Racine carrée de a = ',suite_heron(a):2:4);
end.

```

## Exercice 8 :

```

Program exe8_158;
uses wincrt;
type tab=array[0..100] of real;
var coef:tab;
  x:real;
  i,n:integer;
(***** p.p *****)
function Polynom_horner(n:integer;x:real;coef:tab):real;
var i:integer;
  pol:real;
begin
  pol:=0;
  for i:=n downto 0 do pol:= pol*x + coef[i];
  polynom_horner:= pol;
end;
(***** p.p *****)
begin
  write('Polynôme de degré n = ');readln(n);

  write('X = ');readln(x);

  writeln('Saisir les coefficients du polynome');
  for i:=0 to n do
    begin
      write('a',i,' : ');
      readln(coef[i]);
    end;

  writeln('P(x) = ', polynom_horner(n,x,coef):2:3);
end.

```

## Exercice 1 :

```

Program exe1_192;
uses wincrt;
var ch : string;
(*****)
Procedure lecture (var ch:string);
begin
  repeat
    writeln('Saisir une phrase');
    readln(ch);
    until upcase(ch[1]) in ['A'..'Z'];
end;
(*****)
Procedure compte(ch:string);
var i,nb:integer;
begin
  nb := 0;
  for i:=1 to length(ch) do
    if ch[i]=' '
      then nb:=nb+1;
  writeln('le nombre des mots dans la phrase est : ', nb+1);
end;
(*****p.p*****)
BEGIN
  lecture(ch);
  compte(ch);
END.

```

## Exercice 3 :

```

Program exe3_192;
uses wincrt ;
type tab = array [1..100] of integer ;
var t : tab ;
n : integer ;
(*****)
Procedure lecture(var n:integer;var t:tab);
var i:integer;
begin
  repeat
    writeln('saisir le nombre d"éléments du tableau' );
    readln (n) ;
    until n in [1..100] ;
    writeln ('saisir les ', n, ' éléments de t') ;
    for i:= 1 to n do readln (t[i]) ;
end;
(*****)
Procedure inverse_tout(n:integer;var t:tab);
var i:integer;
{=====}
Procedure permut (var x,y:integer);
var aux:integer;
begin
  aux:=x;
  x:=y;
  y:=aux;
end;
{=====}
begin
  for i := 1 to (n div 2) do
    permut(t[i], t[n-i+1]);
end;

```

```

(*****)
Procedure resultat(n:integer;t:tab);
var i:integer;
begin
  writeln ('tableau inversé :');
  for i := 1 to n do write(t[i]:4) ;
end;
(*****p.p*****)
begin
  lecture(n,t);
  inverse_tout(n,t);
  resultat(n,t);
end.

```

## Exercice 5 :

```

Program exe5_193;
uses wincrt ;
const n_max = 20;
type tab=array [1..n_max] of integer;
var t:tab;
n,x,p : integer;
(*****)
Procedure saisies(var n,x,p:integer; var t:tab);
var i:integer;
begin
  repeat
    writeln ('donner le nombre d"éléments de t');
    readln (n);
    until (n>=1) and (n<n_max);
    writeln ('saisir les éléments de t') ;
    for i:=1 to n do readln (t[i]);
    writeln ('donner l"élément x à insérer');readln (x);
    repeat
      writeln ('donner la position d"insertion'); readln (p)
      until p in [1..n];
end;
(*****)
Procedure insertion(n,x,p:integer;var t:tab);
var i:integer;
begin
  {décalage des éléments vers droite}
  for i:= n downto p do t[i+1] := t[i];

  t[p]:=x; {insertion de l'élément x à la position p}
end;
(*****)
Procedure affiche(n:integer;t:tab);
var i:integer;
begin
  for i:=1 to n+1 do write(t[i]:4);
end;
(*****p.p*****)
begin
  saisies(n,x,p,t);
  insertion(n,x,p,t);
  affiche(n,t);
end.

```